



UNIVERSIDADE ESTADUAL DO MARANHÃO - UEMA
CURSO: ENGENHARIA DA COMPUTAÇÃO
DISCIPLINA: MATEMÁTICA DISCRETA PERÍODO: 2025.2
PROFESSOR: DANIEL GUSMÃO PEREIRA

MARCOS ALEXANDRE ARAÚJO GONÇALVES

CONJECTURA DE COLLATZ EM VISUALG

SÃO LUÍS – MA

2025

1. Introdução

Este trabalho tem como objetivo demonstrar a aplicação da Conjectura de Collatz utilizando a linguagem Portugol no ambiente VisuAlg. A conjectura, proposta por Lothar Collatz em 1937, é um problema matemático simples em aparência, mas ainda sem solução geral comprovada. A implementação em VisuAlg busca ilustrar o comportamento da conjectura e a lógica envolvida no processo iterativo.

2. Definição da Conjectura de Collatz

A Conjectura de Collatz afirma que, ao escolher um número natural “n” e aplicar repetidamente duas regras específicas, o resultado final será sempre 1. As regras são: se o número atual for par, ele deve ser dividido por 2; se for ímpar, multiplica-se o número por 3 e soma-se 1 ao resultado. Esse processo é repetido até que o número alcance o valor 1.

3. Explicação do código

A linguagem utilizada foi o Portugol e o código construído através do programa VisuAlg .

Linha 1 -12:

```
1  Algoritmo "Conjectura Collatz"
2  // Disciplina   : Matemática Discreta
3  // Descrição    : Automatização da verificação da Conjectura de Collatz
4  // Autor(a)     : Marcos Alexandre
5  // Data atual   : 28/10/2025
6
7  Var
8      numero, resultadoFinal: vetor [1..999] de inteiro
9      resultado, numeroinicial: inteiro
10     i, n: inteiro
11     igualUm: logico
12
```

Na linha 7 temos o começo da definição do tipo das variáveis, em que o programa prepara a “mesa de trabalho” para poder iniciar o código:

“numero” e “resultadoFinal” são definidos como um vetor linha indo da posição 1 até 999 e é do tipo inteiro. A posição final é definida até 999 para que o usuário possa testar até grandes números caso queira;

“resultado” é a variável usada para intermediar os valores usados nos cálculos;

“numeroinicial” coloca o valor inicial do número para podermos escrever ele em conjunto do resultado;

“i” serve somente para o funcionamento do para e “n” é o valor até qual o usuário deseja testar;

“igualUm” é uma variável do tipo lógico que define como VERDADEIRO ou FALSO a variável.

Linha 13 - 20:

```
13  Inicio
14      escreval("-----")
15      escreval("Conjectura de Collatz")
16      escreval("-----")
17
18      escreval("Digite até qual número você deseja testar, por favor: ")
19      escreval("O universo analisado será de 1 até o número escolhido.")
20      leia(n)
```

Na linha 13 temos o início do funcionamento do programa. O programa descreve tudo que está dentro da função escreval e também lê um valor para a variável “n” na linha 20. Esse valor “n” que vai ser o universo de trabalho do nosso código.

Linha 22-30:

```
22      // Colocando os números do universo em um vetor
23      para i de 1 ate n faca
24          numero[i] <- i
25      fimpara
26
27      escreval(" ")
28      escreval("Realizando os cálculos, por favor aguarde...")
29      escreval("-----")
30
```

Colocamos os números do universo de trabalho, sendo eles de 1 até “n”.
Exemplo: se o usuário digitar 10, o universo de trabalho será de 1 até 10.

Linha 31 - 58:

```
31 // Aplicação da Conjectura de Collatz
32 para i de 1 ate n faca
33     resultado <- numero[i]
34     numeroinicial <- numero[i]
35
36     escreval(" ")
37     escreval("Realizando a conjectura para o número: ", resultado)
38     escreval("-----")
39
40     enquanto (resultado > 1) faca
41         se (resultado % 2 = 0) entao
42             escreval(">> 0 ", resultado, " é par.")
43
44             escreva(">> Logo, ", resultado, " / 2 resulta em: ")
45             resultado <- resultado div 2
46             escreval(resultado)
47         senao
48             escreval(">> 0 ", resultado, " é ímpar.")
49             escreva(">> Logo, (", resultado, " x 3) + 1 resulta em: ")
50             resultado <- (resultado * 3) + 1
51             escreval(resultado)
52         fimse
53     fimenquanto
54
55     escreval(" ")
56     escreval(">> Sendo ", resultado, " o resultado final da conjectura do número ", numeroinicial)
57     resultadoFinal[i] <- resultado
58 fimpara
```

Na linha 22 utilizamos a função “para” que auto incrementa a variável “i” a cada execução, ou seja, na primeira vez o $i = 1$, na segunda o $i = 2$ e assim até o número “n” definido pelo usuário. A partir disso, pegamos o valor que está presente na posição “i” do vetor e colocamos ele na variável “resultado” e “numeroinicial”. A partir disso, na linha 36 a 38 ele escreve dizendo que está realizando a conjectura de tal número que foi obtido no vetor.

Na linha 40 temos propriamente a prática do cálculo para testar a Conjectura. Começamos elas através da função “enquanto”, enquanto o resultado for maior que 1, ele deve realizar tudo que está entre ele e “fimenquanto”.

Na linha 41 - 45 o se verifica se o resto do “resultado” é igual a zero, se for, ele executará o que está entre “entao” e “senao”. Caso não seja, ele executará tudo entre o “senao” e “fimse”, já que se ele não é para então ele é ímpar. Considerando que seja par, então o novo valor de “resultado” será o valor atual de “resultado”

dividido por 2 através da função “div” (caso seja dividido por / ele não consegue colocar o valor dentro do resultado, já que o VisuAlg divide valores em real por padrão e “resultado” é uma variável do tipo inteiro). Caso seja ímpar, então o novo valor de “resultado” será o produto do valor atual do “resultado” e 3, e 1 somado ao produto. O código realizará esse processo até o “resultado” deixar de ser maior que 1.

Na linha 54 a 58 escreve o resultado final ao ser executado todo o processo e diz qual o número inicial do processo. Com isso, ele coloca o valor do resultado final na posição “i” do vetor “resultadoFinal” para verificação posterior.

Linha 60 - 71:

```
60 // Verificação final
61 igualUm <- verdadeiro
62 para i de 1 ate n faca
63   se (resultadoFinal[i] <> 1) entao
64     igualUm <- falso
65   fimse
66 fimpara
67
68 escreval(" ")
69 escreval("Conclusão: É ", igualUm, " que todos os números até ", n, " resultam em 1.")
70 Fimalgoritmo
71
```

Na linha 61 definimos a variável lógica “igualUm” como VERDADEIRO.

Na linha 62 verificamos se o valor na posição “i” do vetor “resultadoFinal” é diferente de 1, caso seja, então a variável “igualUm” receberá o valor lógico FALSO.

Por fim, na linha 69 damos a conclusão final escrevendo na tela se teve algum número de 1 até “n” que deu resultado final diferente de 1 e finalizamos o código com “Fimalgoritmo”.

4. Conclusão

Através da implementação da Conjectura de Collatz no VisuAlg observamos o seu funcionamento através da lógica computacional. Também compreendemos a importância da computação para a matemática, em que a mesma busca a simplificação e a automatização de processos que seriam considerados impossíveis a nível braçal dentro da matemática.